DEPARTMENT OF THE INTERIOR

U.S. GEOLOGICAL SURVEY

BASIC programs for computing displacements, strains, and tilts from quadrilateral measurements

by

Arvid M. Johnson[1]

and

Rex L. Baum[2]

Open-File Report 87-343

Any use of trade names in this report is for descriptive purposes only and does not imply endorsement by the U.S. Geological Survey. This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards.

[1]University of Cincinnati (M.L. 13)
Cincinnati, Ohio 45221

[2]U.S. Geological Survey
Box 25046, MS 966
Denver, CO 80225

1987

INTRODUCTION

Since 1983, we have been using quadrilaterals defined by survey stakes (fig. 1) to measure displacements, strains, and tilts at the surfaces of landslides. A companion paper will describe the use of quadrilaterals and give derivations of the equations needed to compute displacements, strains, and tilts. This report provides user instructions for, and listings of, BASIC programs that perform the computations. It is assumed that the reader is familiar with MBASIC and CP/M. However, we have tried to make the user instructions complete and step by step so that the average user can run the programs successfully.



Figure 1. Quadrilateral. (a) Isometric view showing survey stakes on a hillside. (b) Plan view. By convention, stake A has the highest elevation and the other stakes are named B, C, and D in clockwise rotation from A. In the field, the slope distance between each pair of stakes and the elevation of each stake is measured. The azimuth of line AC is used to orient the data. Angles $\Theta_1, \ldots, \Theta_8$ are computed from the distance measurements by means of the law of cosines.

Four separate programs are used to handle data obtained from the field measurements: The first, called TEMPDAT.BAS, creates data files; it has a companion program, PRINTDAT.BAS, which prints the contents of the files. The third program, QUADDSP.BAS, computes displacements of two points (corners of a quadrilateral) on a landslide, relative to two points off the landslide (fig. 2). The last program, QUADSTR.BAS, computes the magnitude and direction



Figure 2. Sketch of reference system for computing displacements across a discontinuity or landslide boundary. The x-y plane is horizontal; the origin is at point A and the positive x direction coincides with the direction of line (AD). (a) Displacement component u_1 is parallel to x, u_3 (not shown) is vertical, and u_2 is parallel to y. The angle between the horizontal projection of the total displacement and the x axis is λ. (b) Initial position of point B is given by coordinates x_b, y_b; its final position, B', by coordinates x'_b, y'_b. The projection of angle DAB onto the x-y plane is ϕ; that of angle DAB' is ϕ'.

of the principal stretches[3] of each of four triangles in the quadrilateral (fig. 3) and the tilt (change in dip) and change in dip direction of each triangle (fig. 4).

---

[3]The stretch is the ratio of the final (deformed) length to the initial (undeformed) length of a line.



(a)



(b)

Figure 3. Geometrical relationships between a deformed triangle and its stretch ellipse. (a) Triangle ABC in its final state. Angle Θ_1', is angle CAB; angle Θ_2' is angle BCA. (b) The stretch ellipse. S_1 and S_2 are the principal stretches and are parallel to the x_1 and x_2 coordinates. S_{AB}, S_{AC}, and S_{BC} are the components of stretch parallel to lines AB, AC, and BC respectively. The angle α (= β_{AC}') is the acute angle between line AC and x_1; $\beta_1' = \Theta_1' - \alpha$; $\beta_2' = \pi - \alpha - \Theta_2'$.



Figure 4. Dip direction and dip of triangle ABC; point A has the highest elevation. The positive x direction is due north. Line AB parallels coordinate direction ξ_1; its bearing is ω_1. Line AC parallels coordinate direction ξ_2; its bearing is ω_2. The dip direction parallels coordinate direction ξ_{max}, and its bearing is ω_{max}.

PROGRAM FEATURES

The following section gives a brief description of each program, its input, operation, and output[4]. Examples of program runs with sample printouts follow the descriptions. Program listings and explanations of program units are in the appendix.

TEMPDAT.BAS

In the program, TEMPDAT.BAS, the field measurements are typed in by the user as DATA statements. Instructions for typing the DATA statement are listed in lines 1000-1015 of the program (Appendix). When the program is run, the DATA statements are read, and data belonging to a specified quadrilateral are stored in a sequential file that bears the name of the quadrilateral. The date of measurement, azimuth of line AC, and the slope distances are stored in their original form. The stadia intercepts (elevations) of each corner of the quadrilateral are stored as the elevation differences between point A and the other points.

---

[4]The programs require that a printer is available for use. If a printer is not available, then all LPRINT statements must be changed to PRINT statements.

SAMPLE RUN OF TEMPDAT.BAS

(In all the sample runs, boldface type gives the user response to instructions or prompts given by the computer. Computer prompts are underlined. The symbol <Ret> denotes the return key.)

Insert the diskette with MBASIC in drive A and type

MBASIC

The computer responds by showing version and copyright information on the screen. Remove the diskette in drive A and insert the diskette with the application programs. Typing the command RUN "FILENAME" causes the program to be put into memory and executed.

OK
RUN "TEMPDAT.BAS" <Ret>
WHICH QUAD?
User types in quadrilateral name or number, for example,
Q-2 <Ret>

The computer then prints the quadrilateral names and dates on the screen and prints "FOUND IT!" each time that data belonging to the specified quadrilateral are found. When the trailer (a data set for which the azimuth is greater than 360) is encountered, the file is closed. Control returns to line 5, which prints the prompt:

WHICH QUAD?
Q-7 <Ret>
WHICH QUAD?
"Control C" terminates program operation.

PRINTDAT.BAS

PRINTDAT.BAS makes a tabulated listing of the contents of the data file for each quadrilateral. Because neither PRINTDAT.BAS nor TEMPDAT.BAS sorts the data to put it in chronological sequence, the data sets will appear in the order they were typed into the TEMPDAT.BAS program. PRINTDAT.BAS prints only three places to the right of the decimal point. However, measurements stored in data files created by TEMPDAT.BAS are identical to (have the same number of decimal places as) their counterparts in the DATA statements.

SAMPLE RUN OF PRINTDAT.BAS

OK
RUN "PRINTDAT.BAS" <Ret>
QUAD NUMBER?
Q-7 <Ret>

Computer produces printout listed in figure 5. Program execution terminates and control returns to the user.

OK

```
Q-2
2 AUG 1985-5   1.775  3.016  2.016  3.239  3.007  4.204  0.037  0.665  0.511
12 SEP 1986-5   1.754  3.159  2.029  3.434  2.994  4.435  0.042  0.603  0.424
```

```
Q-7
2 AUG 1985 47   1.738  1.584  1.457  1.886  2.419  2.280  0.088  0.460  0.453
12 SEP 1986 47   1.705  1.607  1.435  1.876  2.415  2.259  0.012  0.391  0.422
```

Figure 5. Sample printouts by PRINTDAT.BAS. The top line gives the quadrilateral number. In each succeeding line, the first column gives the date of measurement; the second column gives the azimuth of line A-C (note sign convention given in the appendix); the next six columns give the slope lengths of lines AB, BC, CD, DA, AC, and BD; and the last three columns give the elevation differences AB, AC, and AD.

QUADDSP.BAS

QUADDSP.BAS reads data from files created by TEMPDAT.BAS and computes three-dimensional displacements of the two quadrilateral stakes on the landslide. The user specifies the dates of the initial (reference) and later measurements and then specifies the "reference line" (the two quadrilateral stakes off the landslide).

SAMPLE RUN OF QUADDSP.BAS

OK
RUN "QUADDSP.BAS" <Ret>
STORED UNDER NAME <QUADDSP.BAS>
PROGRAM FOR COMPUTATION OF DISPLACEMENTS AT SURFACE OF A LANDSLIDE, USING QUADRILATERAL MEASUREMENTS.

TO RUN PROGRAM, TYPE IN QUADRILATERAL NUMBER, IN THE FORM, Q-13 AND PUSH <RETURN>
THEN TYPE IN DATE OF READINGS THAT ARE TO BE USED AS A REFERENCE IN THE FORM, 2 JULY 1983 AND PUSH <RETURN>
FINALLY, TYPE IN DATE OF READINGS THAT ARE TO BE USED TO COMPUTE DEFORMATION, IN THE SAME FORM, AND PUSH <RETURN>
QUADRILATERAL NUMBER?
Q-2 <RET>
REFERENCE LINES ARE: AB; AD; BC; CD
REFERENCE LINE? **CD** <RET>
DATE OF REFERENCE MEASUREMENTS? **2 AUG 1985** <Ret>
DATE OF MEASUREMENTS TO BE USED TO COMPUTE DEFORMATION?
12 SEP 1986 <Ret>

The computer searches file Q-2, retrieves the data, and processes it. After a few seconds, the results (fig. 6) are printed by the printer, and program control returns to line 260 which causes the following to appear:

DATE OF MEASUREMENTS TO BE USED TO COMPUTE DEFORMATION?

```
Q-2          2 AUG 1985    12 SEP 1986
ORIGINAL DATA 1

 1.775  3.016  2.016  3.239  3.007  4.204
 1.754  3.159  2.029  3.434  2.994  4.435

 .0367  .66518  .51062
 .0417199  .60301  .42428
ERRORS IN ANGLES W1,W2,W3=-.132939 -.005792  4.02298E-03
ERRORS IN ANGLES W1,W2,W3= 2.09955  .0937307 -.0722566
*********************************************************

IF W IS NEGATIVE THEN MOVEMENT IS TOWARD THE REF. LINE.
DISPLACEMENTS OF POINT A  : U -PARALLEL TO CD W -NORMAL
V VERTICAL
U .292369      W .0807352      V .08634   DISPL .315361
PLUNGE(DOWNSLOPE)  15.8894
ANGLE BETWEEN CD  AND DISPL. DIR.=  15.437
AZIMUTH OF DISPLACEMENT <NEG. ANGLE IS WEST OF NORTH>-52.5368
*********************************************************

IF W IS NEGATIVE THEN MOVEMENT IS TOWARD THE REF. LINE.
DISPLACEMENTS OF POINT B  : U -PARALLEL TO CD W -NORMAL
V VERTICAL
U .321388      W .0260468      V .0671898   DISPL .329368
PLUNGE(DOWNSLOPE)  11.7707
ANGLE BETWEEN CD  AND DISPL. DIR.=  4.63338
AZIMUTH OF DISPLACEMENT <NEG. ANGLE IS WEST OF NORTH>-68.5728
```

Figure 6. Sample printout by QUADDSP.BAS The top line gives the quadrilateral number, the reference date, and the final date. The next two lines list the slope distances for each date, and the elevation differences are given in the fifth and sixth lines. The lines labeled errors in angles give the error of closure (Laurila, 1983, p. 43). Below the line of stars, the components of displacement, U, W, and V ($U=u_1$, $W=u_2$, $V=u_3$ in fig. 2), the total displacement "DISPL", and the direction of the displacement are listed for points A and B. Note that the plunge is positive if the displacements are directed downward.

The user may respond with a new date (the reference date remains the same) or terminate the session with "Control C."

Had either of the dates been typed incorrectly, the computer could not have retrieved the data. Instead, program execution would have terminated and the following message would have appeared on the monitor:

INPUT PAST END
OK
To return to CP/M from MBASIC, type **SYSTEM**.

QUADSTR.BAS

The program, QUADSTR.BAS, reads data from files created by TEMPDAT.BAS and computes the stretches of each measured line in the quadrilateral. The observed stretches are then used to compute the principal stretches and their orientations relative to one of the diagonals. The principal strains and other useful measures of the deformation are then computed from the principal stretches and the results are printed.

SAMPLE RUN OF QUADSTR.BAS

```
OK
RUN "QUADSTR.BAS" <Ret>
STORED UNDER NAME <QUADSTR.BAS>
PROGRAM FOR COMPUTATION OF DEFORMATION AT SURFACE OF LANDSLIDES, USING
  QUADRILATERAL MEASUREMENTS
PROGRAM WRITTEN 1983 BY A.M. JOHNSON

TO RUN PROGRAM, TYPE IN QUADRILATERAL NUMBER, IN THE FORM, Q-13 AND PUSH
  <RETURN>
THEN TYPE IN DATE OF READINGS THAT ARE TO BE USED AS A REFERENCE IN THE
  FORM, 2 JULY 1983 AND PUSH <RETURN>
FINALLY, TYPE IN DATE OF READINGS THAT ARE TO BE USED TO COMPUTE
  DEFORMATION, IN THE SAME FORM, AND PUSH <RETURN>

IF YOU NEED TO ENTER NEW DATA, USE THE PROGRAM NAMED <TEMPDAT.BAS>
QUADRILATERAL NUMBER?
Q-7 <RET>
DATE OF REFERENCE MEASUREMENTS?
2 AUG 1985 <RET>
DATE OF MEASUREMENTS TO BE USED TO COMPUTE DEFORMATION?
12 SEP 1986 <Ret>
```

Computer searches file Q-7 for the data, retrieves data, and performs computations. The results are printed as shown in figure 7.

```
Q-7          2 AUG 1985    12 SEP 1986
ORIGINAL DATA 1

 1.738  1.584  1.457  1.886  2.419  2.28
 1.705  1.607  1.435  1.876  2.415  2.259

 .0879197  .4598  .4529
 .0124395  .390929  .42184
ERRORS IN ANGLES W1,W2,W3=-.0482689  4.01957E-03  4.00932E-03
ERRORS IN ANGLES W1,W2,W3=-.0959164  7.77959E-03  7.08974E-03

SLOPE STRETCHES
AB  BC  CD  AD  AC  BD
 .981013  1.01452  .984901  .994698  .998347  .990789
*******************************************************

STRAINS IN PERCENTAGES
-1.89875  1.45203 -1.50995 -.530225 -.165355 -.921047
POSITIVE AZIMUTHS ARE EAST OF NORTH, NEGATIVE ARE WEST OF NORTH.
AZIMUTHS OF LINES AB,BC,CD,AD,AC,BD        0
-87.3599 -.110075 -97.4016 -9.17892 -47  39.8386
POSITIVE AZIMUTHS ARE EAST OF NORTH, NEGATIVE ARE WEST OF NORTH.
AZIMUTHS OF LINES AB,BC,CD,AD,AC,BD        1
-87.8487 -1.4372 -96.991 -10.0478 -47  38.5556
                        ***********************

        %S1      %S2      %AREA CHANGE
S1         S2           DAREA
BMAX      GAMMA-MAX     PSI-MAX
                        ************************
TRIANGLE       1
ANGLE BETWEEN A-C AND MAX. STRETCH <NEG ANGLE CLOCKWISE>-43.0838
 1.45521      -1.92619      -.49901
 1.01455       .980738       .99501
 44.0291       .0339035      1.94179

 0            DIP DIRECTION -9.30775      DIP           13.7484

 1            DIP DIRECTION  .427012      DIP           13.6295
ROTATION      -9.73476      CHANGE IN DIP  .118877

TRIANGLE       2
ANGLE BETWEEN B-D AND MAX. STRETCH <NEG.ANGLE CLOCKWISE> 60.8332
 1.98957      -1.77799       .176203
 1.0199        .98222        1.00176
 43.9219       .0376491      2.15612

 0            DIP DIRECTION -8.54109      DIP           13.721

 1            DIP DIRECTION -1.94586      DIP           13.6231
ROTATION      -6.59523      CHANGE IN DIP  .0978783
```

```
TRIANGLE        3
ANGLE BETWEEN A-C AND MAX. STRETCH <NEG ANGLE CLOCKWISE>-8.72621
-.121516        -1.97399        -2.09311
 .998785         .98026          .979069
 44.4637         .0187225       1.0726

 0              DIP DIRECTION -8.49844      DIP             13.8956

 1              DIP DIRECTION -1.69083      DIP             13.1294
ROTATION        -6.80762       CHANGE IN DIP  .766235

TRIANGLE        4
ANGLE BETWEEN B-D AND MAX. STRETCH <NEG.ANGLE CLOCKWISE> 34.3767
-.45296         -1.89945        -2.3438
 .995471         .981006         .976562
 44.5807         .0146378        .838623

 0              DIP DIRECTION -9.19921      DIP             13.8947

 1              DIP DIRECTION  .322945      DIP             13.203
ROTATION        -9.52215       CHANGE IN DIP  .691637

############################################################
```

Figure 7. Sample printout by QUADSTR.BAS The format of the lines above "SLOPE STRETCHES" is identical to that in figure 6. The strains (just below the line of stars) are derived from the stretches by the formula Strain = (Stretch-1)x100. The heading, between two lines of pound signs, identifies each member of the 3x3 array of numbers listed for each triangle. The meaning of each abbreviation follows:

- %S1 = $(S_1-1)\times100$, maximum principal strain x 100
- %S2 = $(S_2-1)\times100$, minimum principal strain x 100
- %AREA CHANGE = $(S_1S_2-1)\times100$ = area strain x 100
- S1, maximum principal stretch
- S2, minimum principal stretch
- DAREA = S_1S_2, ratio of final area to initial area
- BMAX, angle between maximum principal stretch and maximum shear strain.
- GAMMA-MAX, maximum shear strain
- PSI-MAX, angle of maximum shear strain, Tan^{-1}(GAMMA-MAX).

For each triangle, the original dip and dip direction are given in the line that starts 0, the final dip and dip direction are in the line that begins with 1. The quantity labeled "ROTATION" is the change in dip direction; it should not be confused with the rotation of a line that becomes parallel to the principal axis of strain. It is positive if counterclockwise and negative if clockwise. The change in dip is positive if the dip is decreasing. The dip direction follows the sign convention that positive angles are east of north, negative angles are west of north.

PROGRAM LANGUAGE

The programs listed in the appendix are written in MBASIC {BASIC-80, revision 5.21 [CP/M version] © 1977-1981 by Microsoft}. A few differences between MBASIC and ANSI standard BASIC (Bent and Sethares, 1978) are worth noting: (1) MBASIC allows several statements to be typed on one line if they are separated by colons. (2) LET is understood in assignment (i.e., arithmetic) statements. (3) The PRINT USING statement is available in MBASIC for formatting output. (4) Files are treated differently. MBASIC requires that files be OPENED before they can receive output from the program or input data to the program. Files must be CLOSED before program execution terminates. Other differences may exist between MBASIC and other extended versions of BASIC. Users should consult the reference manuals for their BASIC systems.

DISCUSSION

The programs presented in this report enable the user to organize large amounts of quadrilateral data efficiently. Three-dimensional displacements or strain and tilt can be computed quickly and accurately from successive sets of measurements. An error-checking routine in the programs is useful for detecting inaccurate sets of measurements. The programs can be used on a microcomputer; therefore, the computations can be performed away from the office. The possibility of doing the computations in the field means that the data can be analyzed and evaluated there, and that any needed changes or additions to the data collection activities can be implemented immediately.

BIBLIOGRAPHY

Bent, R.J., and Sethares, G.C., 1978, BASIC: an intoduction to computer programming: Monterey, California, Brooks/Cole, 239 p.

Laurila, S.H., 1983, Electronic surveying in practice: New York, John Wiley, 388 p.

APPENDIX A

PROGRAM LISTING

PRINTDAT.BAS

```
10 REM   PROGRAM TO PRINT CONTENTS OF DATA FILES
20 REM   STORED AS <PRINTDAT.BAS>
30 WIDTH LPRINT 100
40 LPRINT:LPRINT
50 INPUT "QUAD NUMBER ? ", Q$
60 OPEN "I", #1,Q$
70 LPRINT Q$
80 IF EOF (1) THEN END
90 INPUT#1,DT$,AZ
100 LPRINT DT$,AZ;
110 FOR J=1 TO 6:INPUT #1,A(J):LPRINT USING "####.###"; A(J);:NEXT J
120 FOR J=1 TO 3: INPUT #1,B(J):NEXT J
130 FOR J=1 TO 3:LPRINT USING "####.###";B(J);:NEXT J:LPRINT
140 GOTO 80
```

PROGRAM LISTING

TEMPDAT.BAS

```
1 PRINT "PROGRAM TO PUT QUADRILATERAL DATA INTO FILES.  CONTAINS DATA"
2 PRINT "FROM ASPEN GROVE LANDSLIDE. STORED UNDER <TEMPDAT.BAS>."
5 PRINT:PRINT:INPUT "WHICH QUAD";FILE$
10 OPEN "O",#1,FILE$
20 READ QD$,DT$,AZ
22 PRINT QD$;DT$;
25 IF AZ>360 THEN CLOSE#1:RESTORE:GOTO 5
30 FOR J=1 TO 6:READ A(J):NEXT J
40 FOR J=0 TO 3:READ B(J):NEXT J
50 IF QD$<>FILE$ THEN 20
55 PRINT:PRINT "FOUND IT!",DT$
60 PRINT#1,DT$;",";AZ;
70 FOR J=1 TO 6:PRINT#1,A(J);:NEXT J
80 FOR J=1 TO 3:C(J)=B(J)-B(0):PRINT#1,C(J);:NEXT J
90 GOTO 20
1000 REM *****  INSTRUCTIONS FOR TYPING DATA LINES  **********
1001 REM FIRST LINE GIVES QUAD. NUMBER, DATE, AND AZIMUTH FROM A TO C
1002 REM PLUS ANGLE IS WEST OF NORTH, MINUS ANGLE IS EAST OF NORTH
1005 DATA Q-18,17 MAY 1984,113
1006 REM NEXT LINE IS SERIES OF SLOPE DISTANCES MEASURED BETWEEN STAKES IN THE
1007 REM SEQUENCE  A-B,B-C,C-D,D-A,A-C,B-D
1010 DATA 5.02,7.741,5.132,8.655,9.396,9.855
1012 REM NEXT LINE IS SERIES OF STADIA ROD INTERCEPTS OBTAINED BY LEVELING
1013 REM IN THE SEQUENCE  A,B,C,D
1015 DATA .15320,.53602,2.64744,2.39918
2262 DATA Q-2,2 AUG 1985,-5
2264 DATA 1.775,3.016,2.016,3.239,3.007,4.204
2266 DATA .8391,.8758,1.50428,1.34972
2538 DATA Q-7,2 AUG 1985,47
2540 DATA 1.738,1.584,1.457,1.886,2.419,2.28
2542 DATA 2.58548,2.6734,3.04528,3.03838
2564 DATA Q-2,12 SEP 1986,-5
2566 DATA 1.754,3.159,2.029,3.434,2.994,4.435
2568 DATA 1.21978,1.26150,1.82279,1.64406
2660 DATA Q-7,12 SEP 1986,47
2662 DATA 1.705,1.607,1.435,1.876,2.415,2.259
2664 DATA 2.73918,2.75162,3.13011,3.16102
10000 DATA Q-99,0 AUG 1999,999
```

PROGRAM LISTING

QUADDSP.BAS

```
100 PRINT "STORED UNDER NAME <QUADDSP.BAS> "
110 PI=3.141592653589796#:WIDTH LPRINT 80
120 DIM BC(2,10),B(2,10),D(6)
130 PRINT "PROGRAM FOR COMPUTATION OF DISPLACEMENTS AT SURFACE OF A"
140 PRINT " LANDSLIDE, USING QUADRILATERAL MEASUREMENTS."
150 PRINT: PRINT "TO RUN PROGRAM, TYPE IN QUADRILATERAL NUMBER, IN THE"
160 PRINT "FORM,  Q-13   AND PUSH <RETURN>"
170 PRINT "THEN TYPE IN DATE OF READINGS THAT ARE TO BE USED AS A REFERENCE"
180 PRINT "IN THE FORM,  2 JULY 1983  AND PUSH <RETURN>"
190 PRINT "FINALLY, TYPE IN DATE OF READINGS THAT ARE TO BE USED TO"
200 PRINT "COMPUTE DEFORMATION, IN THE SAME FORM, AND PUSH <RETURN>"
210 PRINT
220 INPUT "QUADRILATERAL NUMBER";Q$
230 PRINT "REFERENCE LINES ARE: AB; AD; BC; CD."
240 INPUT "REFERENCE LINE";L$
250 INPUT "DATE OF REFERENCE MEASUREMENTS";DAT$(0)
260 INPUT "DATE OF MEASUREMENTS TO BE USED TO COMPUTE DEFORMATION";DAT$(1)
270 LPRINT:LPRINT Q$,DAT$(0),DAT$(1)
280 REM                        ############
290 REM   *** SEARCH FILES FOR SPECIFIED DATA  ***************************
300 KK=0:OPEN "I",#1,Q$
310 INPUT#1,DT$,AZ:I=2
320 IF DT$=DAT$(0) THEN I=0:KK=KK+1
330 IF DT$=DAT$(1) THEN I=1:KK=KK+1
340 QUAD$(I,0)=Q$:DATE$(I,0)=DT$:AZIM(I,0)=AZ
350 FOR J=1 TO 6:INPUT#1, A(I,J):NEXT J
360 FOR J=1 TO 3:INPUT#1, C(I,J):NEXT J
370 IF I=2 THEN 310
380 IF KK=2 THEN 450 ELSE 310
390 REM                        ############
400 REM A(I,J) <I=0 FOR REFERENCE DATE,I=1 FOR LATER DATE,J=1 TO 6> IS SERIES
410 REM OF SLOPE DISTANCES,A-B=1,B-C=2,C-D=3,D-A=4,A-C=5,B-D=6
420 REM C(I,J) <I SAME AS ABOVE> IS SERIES OF DIFFERENCES IN ELEVATION BETWEEN
430 REM STAKES, A-B=1,A-C=2,A-D=3.
440 REM   ***   PRINT DATA   ***********************************************
450 LPRINT "ORIGINAL DATA"I:CLOSE#1
460 FOR I=0 TO 1:LPRINT:FOR J=1 TO 6:LPRINT A(I,J);:NEXT J,I:LPRINT
470 FOR I=0 TO 1:LPRINT:FOR J=1 TO 3:LPRINT C(I,J);:NEXT J,I:LPRINT
480 REM                        ############
490 REM COMPUTE HORIZONTAL DISTANCES BETWEEN STAKES. STORE AS BC(I,J)
500 FOR I=0 TO 1:BC(I,1)=SQR(A(I,1)^2-C(I,1)^2)
510 BC(I,5)=SQR(A(I,5)^2-C(I,2)^2):BC(I,4)=SQR(A(I,4)^2-C(I,3)^2)
520 TP=C(I,1)-C(I,2):BC(I,2)=SQR(A(I,2)^2-TP^2)
530 TP=C(I,3)-C(I,2):BC(I,3)=SQR(A(I,3)^2-TP^2)
540 TP=C(I,1)-C(I,3):BC(I,6)=SQR(A(I,6)^2-TP^2)
550 NEXT I
560 REM                        ############
570 REM   *** CHECK ANGLES IN QUADRILATERAL ****************************
580 FOR I=0 TO 1:FOR J=1 TO 6:D(J)=BC(I,J)
590 NEXT J
600 K=1:FOR J=1 TO 4:JP1=J+1:JJ=J+4:IF J=4 THEN JP1=1:JJ=6
610 IF J=3 THEN JJ=5
620 TP=(D(J)^2+D(JJ)^2-D(JP1)^2)/(2*D(J)*D(JJ))
```

```
630 BETA(K)=ATN(SQR(ABS(1-TP^2))/TP):K=K+1
640 TP=(D(JP1)^2+D(JJ)^2-D(J)^2)/(2*D(JP1)*D(JJ))
650 BETA(K)=ATN(SQR(ABS(1-TP^2))/TP):K=K+1:NEXT J
660 GOSUB 1530
670 W1=-2*PI:FOR K=1 TO 8:W1=W1+BETA(K):NEXT K
680 W2=BETA(1)+BETA(8)-BETA(4)-BETA(5)
690 W3=BETA(6)+BETA(7)-BETA(3)-BETA(2)
700 LPRINT "ERRORS IN ANGLES W1,W2,W3=";W1*180/PI;W2*180/PI;W3*180/PI
710 FOR K=1 TO 8:THETA(I,K)=BETA(K):NEXT K
720 NEXT I
730 REM  *** CHOOSE ROUTINE FOR APPRORIATE REFERENCE LINE  *********
740 IF L$="CD" THEN 1020
750 IF L$="AD" THEN 1210
760 IF L$="BC" THEN 1370
770 REM *** COMPUTATIONS OF DISPLACEMENTS FOR SHEAR ZONES****
780 REM XXD,ZZD ARE ORIGINAL AND XD AND ZD ARE FINAL POSITIONS OF PT D
790 REM  ********  REFERENCE LINE AB  ******************************
800 AN=THETA(0,1)+THETA(0,6):P$="D"
810 XXD=BC(0,4)*COS(AN)
820 ZZD=BC(0,4)*SIN(AN)
830 AN=THETA(1,1)+THETA(1,6)
840 XD=BC(1,4)*COS(AN)
850 ZD=BC(1,4)*SIN(AN)
860 U=XD-XXD:W=ZD-ZZD:V=C(1,3)-C(0,3)
870 DIR=ATN(W/U)
880 DIRAZ=180-AZ-(THETA(0,1)-DIR)*180/PI
890 GOSUB 1580
900 AN=THETA(0,3)+THETA(0,8):P$="C"
910 XXC=BC(0,1)-BC(0,2)*COS(AN)
920 ZZC=BC(0,2)*SIN(AN)
930 AN=THETA(1,3)+THETA(1,8)
940 XC=BC(1,1)-BC(1,2)*COS(AN)
950 ZC=BC(1,2)*SIN(AN)
960 U=XC-XXC:W=ZC-ZZC:V=C(1,2)-C(0,2)
970 DIR=ATN(W/U)
980 DIRAZ=180-AZ-(THETA(0,1)-DIR)*180/PI
990 GOSUB 1580
1000 GOTO 260
1010 REM *********  REFERENCE LINE CD  **************************
1020 AN=THETA(0,4)+THETA(0,7):P$="A"
1030 XXA=BC(0,4)*COS(AN)
1040 ZZA=BC(0,4)*SIN(AN)
1050 AN=THETA(1,4)+THETA(1,7)
1060 XA=BC(1,4)*COS(AN)
1070 ZA=BC(1,4)*SIN(AN)
1080 U=XA-XXA:W=ZA-ZZA:V=C(0,3)-C(1,3)
1090 DIR=ATN(W/U)
1100 DIRAZ=-AZ-(THETA(0,5)-DIR)*180/PI
1110 GOSUB 1580
1120 AN=THETA(0,5)+THETA(0,2):P$="B"
1130 XXB=BC(0,3)-BC(0,2)*COS(AN):ZZB=BC(0,2)*SIN(AN)
1140 AN=THETA(1,5)+THETA(1,2)
1150 XB=BC(1,3)-BC(1,2)*COS(AN):ZB=BC(1,2)*SIN(AN)
```

```
1160 U=XB-XXB:W=ZB-ZZB:V=C(1,1)-C(1,2)-(C(0,1)-C(0,2))
1170 DIR=ATN(W/U):DIRAZ=-AZ-(THETA(0,5)-DIR)*180/PI
1180 GOSUB 1580
1190 GOTO 260
1200 REM *********  REFERENCE LINE AD  *************************
1210 AN=THETA(0,1)+THETA(0,6):P$="B"
1220 XXB=BC(0,1)*COS(AN):ZZB=BC(0,1)*SIN(AN)
1230 AN=THETA(1,1)+THETA(1,6)
1240 XB=BC(1,1)*COS(AN):ZB=BC(0,1)*SIN(AN)
1250 U=XB-XXB:W=ZB-ZZB:V=C(1,1)-C(0,1)
1260 DIR=ATN(W/U):DIRAZ=-AZ+(THETA(0,6)-DIR)*180/PI
1270 GOSUB 1580
1280 AN=THETA(0,4)+THETA(0,7):P$="C"
1290 XXC=BC(0,4)-BC(0,3)*COS(AN):ZZC=BC(0,3)*SIN(AN)
1300 AN=THETA(1,4)+THETA(1,7)
1310 XC=BC(1,4)-BC(1,3)*COS(AN):ZC=BC(1,3)*SIN(AN)
1320 U=XC-XXC:W=ZC-ZZC:V=C(1,2)-C(0,2)
1330 DIR=ATN(W/U):DIRAZ=-AZ+(THETA(0,6)-DIR)*180/PI
1340 GOSUB 1580
1350 GOTO 260
1360 REM *********  REFERENCE LINE BC  *************************
1370 AN=THETA(0,3)+THETA(0,8):P$="A"
1380 XXA=BC(0,1)*COS(AN):ZZA=BC(0,1)*SIN(AN)
1390 AN=THETA(1,3)+THETA(1,8)
1400 XA=BC(1,1)*COS(AN):ZA=BC(1,1)*SIN(AN)
1410 U=XA-XXA:W=ZA-ZZA:V=C(1,1)-C(0,1)
1420 DIR=ATN(W/U):DIRAZ=-AZ+(THETA(0,2)+DIR)*180/PI
1430 GOSUB 1580
1440 AN=THETA(0,2)+THETA(0,5):P$="D"
1450 XXD=BC(0,2)-BC(0,3)*COS(AN):ZZD=BC(0,3)*SIN(AN)
1460 AN=THETA(1,2)+THETA(1,5)
1470 XD=BC(1,2)-BC(1,3)*COS(AN):ZD=BC(1,3)*SIN(AN)
1480 U=XD-XXD:W=ZD-ZZD:V=C(1,3)-C(1,1)-(C(0,3)-C(0,1))
1490 DIR=ATN(W/U):DIRAZ=-AZ+(THETA(0,2)+DIR)*180/PI
1500 GOSUB 1580
1510 GOTO 260
1520 REM *************  SUBROUTINE  ****************************
1530 FOR II=1 TO 8:IF BETA(II)<0 THEN BETA(II)=BETA(II)+PI
1540 NEXT II
1550 RETURN
1560 REM   ########    ##########       ###########      #########
1570 REM   *******************  SUBROUTINE  ************************
1580 LPRINT "*************************************************************"
1590 LPRINT
1600 LPRINT "IF W IS NEGATIVE THEN MOVEMENT IS TOWARD THE REF. LINE."
1610 LPRINT "DISPLACEMENTS OF POINT ";P$;"  : U -PARALLEL TO ";L$;" W -NORMAL"
1620 DISP=SQR(U^2+V^2+W^2)
1630 LPRINT "V VERTICAL":LPRINT "U";U;"     W";W;"     V";V;"  DISPL";DISP
1640 PLG=V/SQR(U^2+W^2):PLG=ATN(PLG)*180/PI:LPRINT "PLUNGE(DOWNSLOPE) ";PLG
1650 LPRINT "ANGLE BETWEEN ";L$;"  AND DISPL. DIR.= ";DIR*180/PI
1660 LPRINT "AZIMUTH OF DISPLACEMENT <NEG. ANGLE IS WEST OF NORTH>";DIRAZ
1670 RETURN
```

PROGRAM LISTING

QUADSTR.BAS

```
100 PRINT "STORED UNDER NAME <QUADSTR.BAS>"
110 PI=3.141592653589796#:WIDTH LPRINT 80
120 PRINT "PROGRAM FOR COMPUTATION OF DEFORMATIONS AT SURFACE OF "
130 PRINT " LANDSLIDES, USING QUADRILATERAL MEASUREMENTS."
140 PRINT " PROGRAM WRITTEN 1983 BY A. M. JOHNSON."
150 PRINT: PRINT "TO RUN PROGRAM, TYPE IN QUADRILATERAL NUMBER, IN THE"
160 PRINT "FORM,  Q-13   AND PUSH <RETURN>"
170 PRINT "THEN TYPE IN DATE OF READINGS THAT ARE TO BE USED AS A REFERENCE"
180 PRINT "IN THE FORM,  2 JULY 1983  AND PUSH <RETURN>"
190 PRINT "FINALLY, TYPE IN DATE OF READINGS THAT ARE TO BE USED TO"
200 PRINT "COMPUTE DEFORMATION, IN THE SAME FORM, AND PUSH <RETURN>"
210 PRINT
220 PRINT "IF YOU NEED TO ENTER NEW DATA USE THE PROGRAM NAMED <TEMPDAT.BAS>."
230 INPUT "QUADRILATERAL NUMBER";Q$
240 INPUT "DATE OF REFERENCE MEASUREMENTS";DAT$(0)
250 INPUT "DATE OF MEASUREMENTS TO BE USED TO COMPUTE DEFORMATION";DAT$(1)
260 LPRINT:LPRINT Q$,DAT$(0),DAT$(1)
270 REM                    ###############
280 REM **** SEARCH FILES FOR SPECIFIED DATA ********************
290 REM                    ###############
300 KK=0:OPEN "I",#1,Q$
310 INPUT#1,DT$,AZ:I=2
320 IF DT$=DAT$(0) THEN I=0:KK=KK+1
330 IF DT$=DAT$(1) THEN I=1:KK=KK+1
340 QUAD$(I,0)=Q$:DATE$(I,0)=DT$:AZIM(I,0)=AZ
350 FOR J=1 TO 6:INPUT#1, A(I,J):NEXT J
360 FOR J=1 TO 3:INPUT#1, C(I,J):NEXT J
370 IF I=2 THEN 310
380 IF KK=2 THEN 460 ELSE 310
390 REM                    ###############
400 REM A(I,J) <I=0 FOR REFERENCE DATE,I=1 FOR LATER DATE,J=1 TO 6> IS SERIES
410 REM OF SLOPE DISTANCES,A-B=1,B-C=2,C-D=3,D-A=4,A-C=5,B-D=6
420 REM C(I,J) <I SAME AS ABOVE> IS SERIES OF DIFFERENCES IN ELEVATION BETWEEN
430 REM STAKES, A-B=1,A-C=2,A-D=3.
440 REM                    ###############
450 REM **********   PRINT DATA   ****************************************
460 LPRINT "ORIGINAL DATA"I:CLOSE#1
470 FOR I=0 TO 1:LPRINT:FOR J=1 TO 6:LPRINT A(I,J);:NEXT J,I:LPRINT
480 FOR I=0 TO 1:LPRINT:FOR J=1 TO 3:LPRINT C(I,J);:NEXT J,I:LPRINT
490 REM                    ###############
500 REM COMPUTE HORIZONTAL DISTANCES BETWEEN STAKES. STORE AS B(I,J)
510 FOR I=0 TO 1:H(I,1)=C(I,1):H(I,4)=C(I,3):H(I,5)=C(I,2)
520 B(I,1)=SQR(A(I,1)^2-C(I,1)^2)
530 B(I,5)=SQR(A(I,5)^2-C(I,2)^2):B(I,4)=SQR(A(I,4)^2-C(I,3)^2)
540 H(I,2)=C(I,2)-C(I,1):B(I,2)=SQR(A(I,2)^2-H(I,2)^2)
550 H(I,3)=C(I,2)-C(I,3):B(I,3)=SQR(A(I,3)^2-H(I,3)^2)
560 H(I,6)=C(I,3)-C(I,1):B(I,6)=SQR(A(I,6)^2-H(I,6)^2)
570 NEXT I
580 REM                    ###############
600 REM******   CHECK ANGLES OF QUADRILATERAL
```

```
610 FOR I=0 TO 1:FOR J=1 TO 6:D(J)=B(I,J):NEXT J
620 K=1:FOR J=1 TO 4:JP1=J+1:JJ=J+4:IF J=4 THEN JP1=1:JJ=6
630 IF J=3 THEN JJ=5
640 TP=(D(J)^2+D(JJ)^2-D(JP1)^2)/(2*D(J)*D(JJ))
650 BETA(K)=ATN(SQR(ABS(1-TP^2))/TP):K=K+1
660 TP=(D(JP1)^2+D(JJ)^2-D(J)^2)/(2*D(JP1)*D(JJ))
670 BETA(K)=ATN(SQR(ABS(1-TP^2))/TP):K=K+1:NEXT J
680 GOSUB 1700
690 W1=-2*PI:FOR K=1 TO 8:W1=W1+BETA(K):NEXT K:C1=W1/8
700 W2=BETA(1)+BETA(8)-BETA(4)-BETA(5):C2=W2/4
710 W3=BETA(6)+BETA(7)-BETA(3)-BETA(2):C3=W3/4
720 LPRINT "ERRORS IN ANGLES W1,W2,W3=";W1*180/PI;W2*180/PI;W3*180/PI
730 FOR K=1 TO 8:BTA(I,K)=BETA(K):NEXT K:REM ****NEW***
740 NEXT I
750 REM              #################
760 REM   *******   COMPUTE STRETCHES OF LINES   ********************
770 LPRINT:LPRINT "SLOPE STRETCHES":LPRINT "AB  BC  CD  AD  AC  BD"
780 FOR J=1 TO 6:S(J)=A(1,J)/A(0,J):LPRINT S(J);:NEXT J
790 LPRINT "*******************************************************"
800 LPRINT:LPRINT "STRAINS IN PERCENTAGES"
810 FOR J=1 TO 6:ST(J)=(S(J)-1)*100:LPRINT ST(J);:NEXT J
820 LPRINT
830 REM              ####################
840 REM  ****    COMPUTE ANGLES OF EACH TRIANGLE IN THE
850 REM  QUADRILATERAL, ANGLES ARE THETA (1) --- THETA (8)  *****************
860 FOR J=1 TO 6:D(J)=A(1,J):NEXT J
870 K=1:FOR J=1 TO 4:JP1=J+1:JJ=J+4:IF J=4 THEN JP1=1:JJ=6
880 IF J=3 THEN JJ=5
890 TP=(D(J)^2+D(JJ)^2-D(JP1)^2)/(2*D(J)*D(JJ))
900 THETA(K)=ATN(SQR(ABS(1-TP^2))/TP):K=K+1
910 TP=(D(JP1)^2+D(JJ)^2-D(J)^2)/(2*D(JP1)*D(JJ))
920 THETA(K)=ATN(SQR(ABS(1-TP^2))/TP):K=K+1:NEXT J
930 REM  **** COMPUTE DIRECTIONS OF LINES IN THE QUADRILATERAL,    ******
940 REM  **** USING AZIMUTH OF LINE AC AS REFERENCE  ********************
950 FOR I=0 TO 1:AM=AZ*PI/180
960 BITA(I,1)=-AM-BTA(I,1):BITA(I,2)=-AM+BTA(I,2):BITA(I,3)=-AM-BTA(I,5)
970 BITA(I,4)=-AM+BTA(I,6):BITA(I,5)=-AM:BITA(I,6)=-AM+BTA(I,2)+BTA(I,3)
980 LPRINT "POSITIVE AZIMUTHS ARE EAST OF NORTH, NEGATIVE ARE WEST OF NORTH."
990 LPRINT "AZIMUTHS OF LINES AB,BC,CD,AD,AC,BD",I
1000 FOR J=1 TO 6:LPRINT BITA(I,J)*180/PI;:NEXT J,I:LPRINT
1010 REM             ######################
1020 LPRINT:LPRINT "        %S1       %S2     %AREA CHANGE"
1030 LPRINT "S1         S2          DAREA"
1040 LPRINT "BMAX      GAMMA-MAX     PSI-MAX"
1050 REM             ####################
1060 REM ***  COMPUTE PRINCIPAL STRETCHES AND DIRECTIONS ******
1070 FOR J=1 TO 4:JJ=J+4:IF J=3 THEN JJ=5
1080 LPRINT "TRIANGLE",J
1090 JP1=J+1:IF J=4 THEN JJ=6:JP1=1
1100 T0=1/S(J)^2-1/S(JP1)^2
```

```
1110 T1=T0/(1/S(J)^2-1/S(JJ)^2)
1120 IF J=1 THEN AG(5)=0:AG(1)=-THETA(1):AG(2)=THETA(2)-PI:GOTO 1160
1130 IF J=2 THEN AG(6)=0:AG(2)=-THETA(3):AG(3)=THETA(4)-PI:GOTO 1160
1140 IF J=3 THEN AG(5)=0:AG(4)=THETA(6):AG(3)=PI-THETA(5):GOTO 1160
1150 IF J=4 THEN AG(6)=0:AG(1)=THETA(8):AG(4)=PI-THETA(7):GOTO 1160
1160 T2=COS(2*AG(J))-COS(2*AG(JJ)):T3=COS(2*AG(J))-COS(2*AG(JP1))
1170 TP=T1*T2-T3
1180 T4=SIN(2*AG(J))-SIN(2*AG(JJ)):T5=SIN(2*AG(J))-SIN(2*AG(JP1))
1190 TP=TP/(T1*T4-T5):ALPHA(J)=(1/2)*ATN(TP)
1200 TV1=2*T0/(COS(2*ALPHA(J))*T3-SIN(2*ALPHA(J))*T5)
1210 TV2=2*(1/S(J)^2)-TV1*COS(2*(ALPHA(J)+AG(J)))
1220 T1=(1/2)*(TV1+TV2):T2=(1/2)*(TV2-TV1)
1230 REM *** S1=MAX. PRINCIPAL STRETCH; S2=MIN. PRINCIPAL STRETCH
1240 S1=SQR(1/T1):S2=SQR(1/T2)
1250 IF S2>S1 THEN ALPHA(J)=ALPHA(J)+PI/2:GOTO 1200
1260 AL1=ALPHA(J)*180/PI
1270 BMAX=ATN(S2/S1)
1280 GAMMAMAX=((S1/S2)-(S2/S1))/2:PSIMAX=ATN(GAMMAMAX)
1290 REM***  BMAX IS THE DIRECTION OF MAX. SHEAR STRAIN MEASURED FROM S1.
1300 REM***  GAMMAMAX IS THE MAXIMUM SHEAR STRAIN AMD PSIMAX IS THE ANGLE
1310 REM*** OF MAX SHEAR STRAIN.
1320 IF JJ=5 THEN 1330 ELSE 1350
1330 LPRINT "ANGLE BETWEEN A-C AND MAX. STRETCH <NEG ANGLE CLOCKWISE>";AL1
1340 GOTO 1360
1350 LPRINT "ANGLE BETWEEN B-D AND MAX. STRETCH <NEG.ANGLE CLOCKWISE>";AL1
1360 LPRINT (S1-1)*100,(S2-1)*100,(S1*S2-1)*100
1370 LPRINT S1,S2,S1*S2
1380 LPRINT BMAX*180/PI,GAMMAMAX,PSIMAX*180/PI
1390 REM                    ###############
1400 REM  *********   COMPUTE DIPS AND DIP DIRECTIONS  ******************
1410 FOR I=0 TO 1:DZDX1=H(I,J)/B(I,J):DZDX2=H(I,JJ)/B(I,JJ)
1420 DXDX1=COS(BITA(I,J)):DYDX1=SIN(BITA(I,J)):DXDX2=COS(BITA(I,JJ))
1430 DYDX2=SIN(BITA(I,JJ))
1440 IF ABS(DXDX2)<1E-10 THEN SWAP DZDX1,DZDX2:SWAP DXDX1,DXDX2:SWAP DYDX1,DYDX2
1450 R=DXDX2/DXDX1:BS=(DZDX2-R*DZDX1)/(DYDX2-R*DYDX1)
1460 AS=(DZDX2-BS*DYDX2)/(DXDX2):DIPDIR(I)=ATN(BS/AS)
1470 DIP(I)=AS*COS(DIPDIR(I))+BS*SIN(DIPDIR(I)):DIP(I)=ATN(DIP(I))
1480 IF DIP(I)<0 THEN DIPDIR(I)=DIPDIR(I)+PI:DIP(I)=-DIP(I)
1490 LPRINT:LPRINT I,"DIP DIRECTION",DIPDIR(I)*180/PI,"DIP",DIP(I)*180/PI
1500 NEXT I
1510 LPRINT "ROTATION",(DIPDIR(0)-DIPDIR(1))*180/PI,
1520 LPRINT "CHANGE IN DIP",(DIP(0)-DIP(1))*180/PI:LPRINT
1530 NEXT J
1540 LPRINT "******************************************************************"
1550 LPRINT
1560 GOTO 240
1570 REM              ######################
1690 REM *********** SUBROUTINE *********************
1700 FOR II=1 TO 8:IF BETA(II)<0 THEN BETA(II)=BETA(II)+PI
1710 NEXT II
1720 RETURN
```